

Hans Walser, [20071003a]

Die Klothoide

Es wird ein Programm (MuPAD) vorgestellt, mit welchem jede durch die so genannte natürliche Gleichung gegebene Kurve approximativ gezeichnet werden kann. Insbesondere kann damit auch die Klothoide gezeichnet werden.

1 Kurvenkrümmung

Unter der Krümmung $\kappa(s)$ verstehen wir die momentane Richtungsänderung der Kurve beim Parameterwert s . Dabei soll s die Kurvenlänge sein (so genannter *natürlicher Parameter*). Das Integral

$$\int_{s_0}^{s_1} \kappa(s) ds$$

gibt die gesamte Richtungsänderung beim Durchlauf des Intervalls $[s_0, s_1]$. Der Kehrwert von $|\kappa(s)|$ ist der lokale Krümmungskreisradius.

2 Die natürliche Gleichung einer Kurve

Die Gleichung $\kappa = \kappa(s)$ legt eine Kurve im Wesentlichen, das heißt bis auf Anfangspunkt und Anfangsrichtung fest. Diese Gleichung heißt deshalb die *natürliche Gleichung* der Kurve.

Wir arbeiten in der komplexen Zahlenebene und bezeichnen den Anfangspunkt mit a und die Anfangsrichtung mit ϕ ,

3 Diskretisation durch eine Punktfolge

Wir wählen eine Schrittlänge Δs und definieren eine Folge $\{z_j\}$ komplexer Zahlen mit den Startwerten $z_0 = p$ und $z_1 = z_0 + \Delta s e^{i\phi}$ rekursiv:

$$z_{j+1} = z_j + (z_j - z_{j-1}) e^{i\Delta s \kappa(j\Delta s)}$$

Wir verbinden die Punkte dieser Folge mit einem Polygonzug; das gibt eine Approximation der gesuchten Kurve.

Beispiel: Die konstante Krümmung $\kappa(s) = 2$ gibt einen Kreis mit dem Radius $\frac{1}{2}$.

```
Laenge:=2:
p:=1+1/2*I:
phi:=PI/6:
deltas:=0.01:

kappa:=s->2:

n:=ceil(Laenge/deltas):

z[0]:=p:
z[1]:=z[0]+deltas*exp(I*phi):
```

```


for j from 1 to n-1 do
  z[j+1]:=z[j]+(z[j]-z[j-1])*exp(I*deltas*kappa(j*deltas));
end_for:

Polygon:=i->plot::Polygon2d([[Re(z[i]), Im(z[i])],
[Re(z[i+1]), Im(z[i+1])]], LineWidth=0.5,
LineColor=[1,0,0]):

Ursprung:=plot::Point2d([0,0], PointSize=2,
PointColor=[0,0,0]):

plot(Ursprung, Polygon(i)$i=0..n-1, Scaling=Constrained,
TicksDistance=1, TicksBetween=0, Width=60, Height=60);

```



Kreis

4 Die Klothoide

Die *Klothoide*, auch *Spinnkurve* oder CORNUSche Spirale genannt, ist eine Kurve, deren Krümmung proportional zur Bogenlänge ist.

Die Klothoide hat also eine natürliche Gleichung von der Form $\kappa(s) = \frac{1}{a^2}s$. Die Schreibweise $\frac{1}{a^2}$ des Proportionalitätsfaktors hat historische Gründe. Die Klothoide wird im Bahn- und Straßenbau verwendet. Bei konstanter Durchfahrgeschwindigkeit, also bei $|\dot{x}| = \text{const.}$ ist dann die Radialbeschleunigung proportional zur Bogenlänge und damit zur Fahrzeit. Die Berechnung der Klothoide führt allerdings auf Integrale, die nicht elementar auswertbar sind. Man kann zeigen, dass der „Wickelpunkt“ A die Koordinaten $A\left(\frac{a\sqrt{\pi}}{2}, \frac{a\sqrt{\pi}}{2}\right)$ hat.


```

Laenge:=20:
p:=0:
phi:=0:
deltas:=0.01:

a:=2:
kappa:=s->s/a^2:

```

```
n:=ceil(Laenge/deltas):  
  
z[0]:=p:  
z[1]:=z[0]+deltas*exp(I*phi):  
  
for j from 1 to n-1 do  
  z[j+1]:=z[j]+(z[j]-z[j-1])*exp(I*deltas*kappa(j*deltas));  
end_for:  
  
Polygon:=i->plot::Polygon2d([[Re(z[i]), Im(z[i])],  
[Re(z[i+1]), Im(z[i+1])]], LineWidth=0.5,  
LineColor=[1,0,0]):  
  
Wickelpunkt:=plot::Point2d([a*sqrt(PI)/2,a*sqrt(PI)/2],  
PointSize=3, PointColor=[0,0,1]):  
  
Ursprung:=plot::Point2d([0,0], PointSize=2,  
PointColor=[0,0,0]):  
  
plot(Ursprung, Polygon(i)$i=0..n-1, Wickelpunkt,  
Scaling=Constrained, TicksDistance=1, TicksBetween=0,  
Width=100, Height=100);
```

**Klothoide mit Wickelpunkt**